```
        .data
        #A, column-major order
mat_A: .word 0, 4, 8, 12, 1, 5, 9, 13, 2, 6, 10, 14, 3, 7, 11, 15
        #B: column-major order
mat_B: .word 0, 4, 8, 12, 1, 5, 9, 13, 2, 6, 10, 14, 3, 7, 11, 15
        #C: row-major order
mat_C: .word 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

M:      .word 4 #M=4
N:      .word 4 #N=4
K:      .word 4 #K=4
white: .asciiz "␣"
newL:  .asciiz "\n"

        .text
main:   lw $s0, M($0)
        lw $s1, N($0)
        lw $s2, K($0)

        li $t3, 0 #i=0
        li $t4, 0 #j=0
        li $t5, 0 #z=0

loopM:

loopN:
        mult $t3, $s1
        mflo $t6
        add  $t6, $t6, $t4   #C: i * N + j ($t6)
        sll  $t6, $t6, 2
        li   $t2, 0    #Acumulador productos (fila x columna)
loopK:
        mult $t5, $s0
        mflo $t7
        add  $t7, $t7, $t3   #A: z * M + i ($t7)
        sll  $t7, $t7, 2

        mult $t4, $s2
        mflo $t8
        add  $t8, $t8, $t5   #B: j * K + z ($t8)
        sll  $t8, $t8, 2
```

```
        lw   $t0, mat_A($t7) #Carga A
        lw   $t1, mat_B($t8) #Carga B
        mult $t0, $t1        #A[z * M + i] * B[j * K + z]
        mflo $t0
        add  $t2, $t2, $t0

        addi $t5, $t5, 1     #z++;
        bne  $t5, $s2, loopK
        li   $t5, 0
        sw   $t2, mat_C($t6) #Almacenar fila x columna en C ($t6)
        addi $t4, $t4, 1     #j++;
        bne  $t4, $s1, loopN
        li   $t4, 0
        addi $t3, $t3, 1     #i++;
        bne  $t3, $s0, loopM

        mult $s0, $s1   #Muestra los resultados. C-row-major
        mflo $s2
        li   $t0, 0
        li   $t1, 0
        li   $t3, 0

loopP:  lw $t2, mat_C($t0)
        addi $t0, $t0, 4

        move $a0, $t2       #I/O syscall: code 1, mostrar entero
        li   $v0, 1
        syscall

        la $a0, white       #I/O syscall: code 4, mostrar string
        li $v0, 4
        syscall

        addi $t3, $t3, 1
        bne  $t3, $s0, noNewL
        li   $t3, 0

        la $a0, newL        #I/O syscall: code 4, mostrar string
        li $v0, 4
        syscall

noNewL: addi $t1, $t1, 1
        bne  $t1, $s2, loopP

end:
        li $v0, 10          #I/O syscall: code 10, salida
        syscall
```